

## Affichage par PLOT d'une chaîne de caractères avec attributs

par André C.

Avec le Super-Oric, les commandes d'affichage :

```
PRINTéléments_à_imprimer,  
PRINT@,x,y;éléments_à_imprimer  
PLOTx,y,chaîne_de_caractères
```

affichent selon les couleurs PAPER et INK en cours. Mais comme avec l'Oric classique, ces commandes PAPER et INK modifient la couleur de tous les éléments de l'écran texte, même ceux qui sont déjà affichés.


Avec l'Oric classique, pour changer spécifiquement les couleurs d'un élément, il faut placer par-devant un ou plusieurs attributs.

Avec le Super-Oric, les attributs sont intégrés au code ASCII de chaque caractère. En conséquence, il faut un "word" (2 octets) pour définir chaque caractère. Seule la commande **PLOT x,y,word** permet d'afficher les caractères avec leurs attributs.

Le programme ci-dessous illustre la différence entre PRINT et PLOT. Il fait appel à la routine universelle "Affichage par PLOT d'une chaîne de caractères avec attributs" (lignes 22000 à 22 70).



```
100 \ Test du S/P d'affichage  
110 \ Code PA pour No de Palette, priorite etc.  
120 GOSUB 33001: \ Redefinition des couleurs  
130 CLS: \ Ecran blanc  
140 PRINT@10,8;"SALUT"  
150 S$="LES GARS":PA=#2400:X=10:Y=10:GOSUB 22020  
160 S$="A PLUS...":PA=#2800:X=10:Y=12:GOSUB 22020  
170 END  
22000 \ Affichage d'une chaine de caracteres avec attribut  
22010 \ En entrée S$, PA, X et Y  
22020 X=X-1:L=LEN(S$)  
22030 FOR I=1 TO L  
22040 C$=MID$(S$,I,1):AS=ASC(C$)  
22050 PLOT X+I,Y,AS+PA  
22060 NEXT  
22070 RETURN  
33000 \ Redefinition des couleurs Oric  
33001 DEF INK 0,#7FFF: \Palette 0 Couleur 0 Blanc (Oric 7)  
33002 DEF INK 1,#0000: \Palette 0 Couleur 1 Noir (Oric 0)  
33003 DEF INK 2,#7FE0: \Palette 0 Couleur 2 Cyan (Oric 6)  
33004 DEF INK 3,#03E0: \Palette 0 Couleur 3 Vert (Oric 2)  
33005 DEF INK 4,#7FFF: \Palette 1 Couleur 0 Blanc (Oric 7)  
33006 DEF INK 5,#7C00: \Palette 1 Couleur 1 Bleu (Oric 4)  
33007 DEF INK 6,#03FF: \Palette 1 Couleur 2 Jaune (Oric 3)  
33008 DEF INK 7,#7C1F: \Palette 1 Couleur 3 Magenta (Oric 5)  
33009 DEF INK 8,#03E0: \Palette 2 Couleur 0 Vert (Oric 2)  
33010 DEF INK 9,#001F: \Palette 2 Couleur 1 Rouge (Oric 1)  
33011 RETURN
```



On observe que le code PA est ajouté au code Ascii. Ici PA indique que les caractères doivent être affichés avec la priorité 1 (PA=#2xxx) et la palette 1 (PA=#2400) ou 2 (PA=#2800).